


Police Patrol



This game hiresroutine is based on the hiresroutine of Othello, but then used for a 7x7 screen. To give it a more hiresvalue the cardata is changed every line and so creating a larger hires screen than actually available.

The 100 levels are compressed in as many ways possible. Overlapping positions between levels are used and x/y-positions are stored in 1 byte. The remaining bits are used to store a next levelindicator and the not used fields counter.

the source:

```
; ; Police Patrol, ZXPAND support

max          EQU  100          ; 100 in game

? * TORNADO *

          ORG  #4009
          DUMP 49161

nought     EQU  28

          JP   begin
d_file     DEFW dfile
dfcc       DEFW dfile+1
var        DEFW vars
dest       DEFW 0
eline      DEFW last
chadd      DEFW last-1
xptr       DEFW 0
stkbot     DEFW last
stkend     DEFW last
berg       DEFB 0
mem        DEFW 0          ; was membot, now used as LBUF
          DEFB 0
dfsz       DEFB 2
stop       DEFW 1
lastk      DEFB 255,255,255
margin     DEFB 55
nxtlin     DEFW basic
oldppc     DEFW 0
blkcnt     DEFB 0
;flagx     DEFB 0
strlen     DEFW 0
taddr      DEFW 3213
seed       DEFW 0
frames     DEFW 65535
coords     DEFB 0,0
prcc       DEFB 188
```

```

sposn      DEFB 33,24
cdflag     DEFB 64

waitkey    PUSH DE
           LD  D,8
           LD  HL,lastk
wkey       LD  A,255
           LD  (HL),A
           LD  BC,%1110000000000111
           LD  A,#A0
           OUT (C),A
           EX  (SP),HL
           EX  (SP),HL
           IN  A,(C)
           CPL
           AND 31
           JR  Z,keybkey
           LD  HL,keys-1
fzxkey     INC  HL
           ADD  A,A
           JR  NC,fzxkey
           LD  A,(HL)
           POP DE
           RET

keybkey    DEC  A
wkey0      CP  (HL)
           JR  Z,wkey0
           DEC D
           JR  NZ,wkey
           LD  BC,(lastk)
           CALL #7BD
           POP DE
           RET

line       DEFB 255,255,255,255,255,255,255
           DEFB 255,255,255,255,255,255,255

nlevel     LD  HL,level+1
           LD  A,(HL)
levell     INC  A ; check on last
           LD  (HL),A
           SUB max+1
           JR  Z,levell

begin      LD  IX,hr
restart    LD  B,49
           LD  SP,#43FF
           LD  HL,#0180 ; needed, error when intrupt
           LD  (moveon+1),HL ; while in cls-routine
           LD  HL,screen
cls        LD  (HL),128
           INC HL
           LD  (HL),1
           INC HL
           DJNZ cls

level      LD  A,0 ; 0 1 for demo
           OR  A
           LD  DE,#400
           JR  NZ,ingame
           LD  HL,levelcode
rlevel     PUSH HL
           CALL waitkey
           LD  HL,#7D ; keycode to 'ascii'
           ADD A,L
           LD  L,A
           LD  A,(HL)
           POP HL
           LD  (HL),A ; code on screen
           INC HL
           DEC D
           JR  NZ,rlevel
           LD  HL,levelcode
           LD  A,(HL)
           SUB 28

```

```

LD D,A
ADD A,A
ADD A,A
ADD A,D
ADD A,A
INC HL
ADD A,(HL) ; input back to number
SUB 28
INC HL
INC A
CP max+1
JR NC,nlevel ; number too high
LD E,A
LD A,(HL)
SUB 28
ADD A,A
ADD A,A
ADD A,A
ADD A,A
INC HL
ADD A,(HL)
SUB 28

LD D,A
LD A,E
ingame LD B,A
LD HL,leveldata-1
flevel INC HL
BIT 7,(HL)
JR Z,flevel
DJNZ flevel
LD A,E
OR A
JR Z,nblock-3 ; during game no check
; check firstfield with code
LD A,(HL)
AND #77
CP D ; compare code
JR NZ,nlevel ; false code goes to 1
LD A,E
LD (level+1),A ; set levelnumber
LD BC,#300+nought
nblock XOR A ; now set leveldata
LD E,(HL)
RLD
AND 7
LD D,A
RLD
BIT 3,A
JR Z,nocnt
nocnt INC C
AND 7
LD (HL),E ; restore data
LD E,A
INC HL
PUSH HL
CALL field
LD (HL),255
INC HL
LD (HL),255
POP HL
DJNZ nblock
DEC HL
DEC HL
DEC HL
LD A,(HL)
AND 7
ADD A,28
LD D,A
LD A,(HL)
RRCA
RRCA
RRCA
RRCA
AND 7
ADD A,28
LD E,A
LD (levelcode+2),DE
LD A,(level+1)
DEC A

```

```

find10    LD    E,255
          INC    E
          LD    D,A
          SUB    10
          JR    NC,find10
          LD    HL,#1C1C
          ADD    HL,DE
          LD    (levelcode),HL    ; set current levelcode
          LD    A,C
          LD    (blkcnt),A
          LD    (blknot),A
          LD    D,H
          LD    DE,255
fstart    INC    E
          CALL   field
          JR    NC,fstart
          LD    A,7
          SUB    D
          LD    (ycar+1),A
          LD    (xcar+1),HL
          DEFB   62                ; hide pop de=false move
undo       POP    DE
gameloop   CALL   waitkey
          CP     13                ; "r"
          JP     Z,restart        ; restart level
          LD     B,5
          LD     HL,keys
fkey       CP     (HL)
          JR     Z,keyfnd
          INC    HL
          DJNZ   fkey
          XOR    A
setjp      JR     Z,gameloop      ; false key

setblock   LD     HL,frames      ; during delay
          LD     C,4              ; screen is refreshed
delay0     LD     A,(HL)         ; and current field
delay1     CP     (HL)         ; blocked by hiresroutine
          JR     Z,delay1
          DEC    C
          JR     NZ,delay0

          LD     HL,blkcnt
          INC    (HL)
vg1        LD     A,(HL)
          CP     46+nought
          JP     Z,nlevel
keyfnd     LD     C,B
          PUSH   DE              ; save old
          DEC    C
          JR     Z,select
          DEC    C
          JR     Z,right
          DEC    C
          JR     Z,left
          DEC    C
          JR     Z,down
          JR     up

select     LD     HL,#ABAA
          LD     (moveon+1),HL    ; set moveon
movein     LD     A,6
          CP     E
          JR     C,undo
          CP     D
          JR     C,undo
          CALL   field
          JR     NC,undo
          POP    AF              ; drop old cursor

          LD     A,7
          SUB    D
          LD     (ycar+1),A
          LD     (xcar+1),HL    ; in hr cardisplay
          LD     A,(moveon+2)
          DEC    A
          JR     setjp

lbuf       PUSH   HL
          EX     (SP),HL

```

```

POP    HL
PUSH   HL
POP    HL
LD     R,A
keys   DEFB 10,5,26,25,1      ; qaopz
      DEFB 62
right  INC  E
      DEFB 62
left   DEC  E
      DEFB 62
up     DEC  D
      DEFB 62
down   INC  D
      DEFB 62
      RET                      ; for lbuf only
      JR    movein

leveldata DEFB #99             ; 0011 1
      DEFB #C0,#43,#33        ; 0140 0
      DEFB #8E                 ; 0206 3
      DEFB #AE,#59,#1A        ; 0326 3
      DEFB #9A                 ; 0412 3
      DEFB #35+#88,#6B,#34+8  ; 0535 3
      DEFB #95                 ; 0615 2
      DEFB #42+#88,#6C,#21+8  ; 0742 3
      DEFB #99                 ; 0811 3
      DEFB #65+#88,#66+8,#69  ; 0965 3
      DEFB #81                 ; 1001 0
      DEFB #80,#44,#59        ; 1100 1
      DEFB #64+#88            ; 1264 3
      DEFB #88,#41+8,#66+8    ; 1300 3
      DEFB #83                 ; 1403 2
      DEFB #89,#0A,#48        ; 1501 3
      DEFB #83+8              ; 1603 3
      DEFB #89,#0E,#0C        ; 1701 3
      DEFB #AD                 ; 1825 3
      DEFB #89,#1B,#33        ; 1901 2
      DEFB #A5                 ; 2025 1
      DEFB #8A,#23,#31        ; 2102 0
      DEFB #D3                 ; 2253 1
      DEFB #84,#3B,#66        ; 2304 1
      DEFB #B1                 ; 2431 2
      DEFB #8E,#29,#48        ; 2506 3
      DEFB #B2                 ; 2632 1
      DEFB #86,#3B,#43        ; 2701 1
      DEFB #95                 ; 2815 2
      DEFB #98,#19,#40        ; 2910 2
      DEFB #C2                 ; 3042 1
      DEFB #99,#25,#55        ; 3111 1
      DEFB #C2                 ; 3132 2
      DEFB #9A,#1D,#5A        ; 3312 3
      DEFB #D8                 ; 3450 3
      DEFB #9B,#28,#31        ; 3513 2
      DEFB #A3                 ; 3623 1
      DEFB #DA,#13,#2C        ; 3752 2
      DEFB #C2                 ; 3832 2
      DEFB #AE,#1C,#59        ; 3926 3
      DEFB #A1                 ; 4021 1
      DEFB #96,#4A,#3B        ; 4116 2
      DEFB #DA                 ; 4252 3
      DEFB #A9,#2B,#5E        ; 4321 3
      DEFB #B3                 ; 4433 1
      DEFB #C4,#69,#3D        ; 4544 2
      DEFB #CB                 ; 4643 3
      DEFB #D9,#6D,#4C        ; 4751 3
      DEFB #D5                 ; 4855 1
      DEFB #CC,#54,#62        ; 4944 1
      DEFB #8A                 ; 5002 3
      DEFB #9B,#4D,#11        ; 5113 2
      DEFB #94                 ; 5214 2
      DEFB #CD,#59,#21        ; 5345 2
      DEFB #92                 ; 5412 1
      DEFB #B8,#31,#24        ; 5530 1
      DEFB #B3                 ; 5633 2
      DEFB #88,#1C,#68        ; 5700 3
      DEFB #C2,#14,#22        ; 5842 0
      DEFB #9C,#2A,#45        ; 5914 2
      DEFB #D1,#25,#26        ; 6051 0
      DEFB #B5,#41,#51        ; 6135 0
      DEFB #E0,#12            ; 6260 1

```

```

DEFB #B9,#1A,#6B      ; 6312 3
DEFB #B6                ; 6436 2
DEFB #DC,#62,#2A       ; 6554 2
DEFB #88,#48            ; 6600 3
DEFB #8A,#6B            ; 6702 3
DEFB #A9,#6E            ; 6821 2
DEFB #DC,#2D            ; 6954 3
DEFB #AB,#26            ; 7023 0
DEFB #B4,#38            ; 7134 2
DEFB #C8,#68            ; 7240 3
DEFB #D9,#61            ; 7351 2
DEFB #A8,#5D            ; 7420 2
DEFB #A3,#03            ; 7523 1
DEFB #CA,#5A,#0D        ; 7642 3
DEFB #85,#13            ; 7705 0
DEFB #A4,#00            ; 7824 0
DEFB #83,#22            ; 7903 0
DEFB #86,#25            ; 8006 0
DEFB #D3,#61            ; 8153 0
DEFB #C2,#0A            ; 8242 1
DEFB #C0,#52            ; 8340 0
DEFB #A4,#28            ; 8424 1
DEFB #C3,#04            ; 8543 0
DEFB #D3,#32            ; 8653 0
DEFB #95,#49            ; 8715 1
DEFB #C6,#23            ; 8846 0
DEFB #A2,#01            ; 8922 0
DEFB #93,#66            ; 9013 0
DEFB #E5,#1D            ; 9165 1
DEFB #A5,#2A            ; 9225 1
DEFB #D3,#0A            ; 9353 1
DEFB #B3,#04            ; 9433 0
DEFB #A1,#20            ; 9521 0
DEFB #B1,#0A            ; 9631 1
DEFB #94,#4D            ; 9714 1
DEFB #A6,#3C            ; 9826 1
DEFB #93,#52,#62        ; 9913 0

```

```

field      LD      HL,screen
           LD      A,D
           ADD     A,A          ; x2
           ADD     A,D          ; x3
           ADD     A,A          ; x6
           ADD     A,D          ; x7
           ADD     A,A          ; x14
           ADD     A,E
           ADD     A,E
           ADD     A,L
fldok      LD      L,A
           LD      A,(HL)
           CP      129
           RET

carudg     DEFB 192,3,130,1,133,1
           DEFB 135,1,159,129,162,65 ,162,33
           DEFB 191,249,191,253,164,37,170,85
           DEFB 132,33,128,1,192,3

hr
hr00       LD      B,4
           DJNZ   hr00
           LD      HL,carudg
           EXX
           LD      DE,14
           LD      HL,screen
           LD      B,7
hr1        LD      A,line/256
           LD      I,A
           LD      A,line*256/256
           LD      C,14
           CALL   lbuf+#8000
           PUSH   HL
           POP    HL

hr2        LD      A,H
           LD      I,A
           LD      I,A
           LD      A,B

```

```

ycar      CP      7
          EXX
xcar      LD      DE,screen
          JR      NZ,delaycar
          LDI
          LDI
carback   EXX
          LD      A,L
          CALL   lbuf+#8000+5
          LD      A,0
          DEC    C
          JP     NZ,hr2
          INC    C
          DEC    BC
          LD     A,line/256
          LD     I,A
          LD     A,line*256/256
          CALL   lbuf+#8000

          ADD    HL,DE
          DJNZ   hr1

hr3       EXX
          EX     (SP),HL
          EX     (SP),HL
moveon    LD      DE,#0180
          INC    DE
          DEC    DE
          LD     HL,(xcar+1)
          LD     (HL),E
          INC    HL
          LD     (HL),D
          LD     BC,#808
          LD     HL,dfile+#8000
          LD     A,#1E
          LD     I,A
          LD     A,#F5
          CALL   #2B5

          CALL   #292
          CALL   #220
          LD     IX,hr
          JP     #2A4

delaycar  IN      A,(1)
          NOP
          JR     carback

dfile     DEFB    118,118,118
          DEFB    "P"-27,"O"-27,"L"-27,"I"-27,"C"-27,"E"-27,0,0
          DEFB    "P"-27,"A"-27,"T"-27,"R"-27,"O"-27,"L"-27,118

          DEFB    118,"L"-27,"E"-27,"V"-27,"E"-27,"L"-27,"C"-27
          DEFB    "O"-27,"D"-27,"E"-27,0
levelcode DEFB    0,0,0,0,118,118

blknot    DEFB    0,118,118,118

          DEFB    0                      ; filler for right screenstart

screen    EQU     $
basic     DEFB    0,1
          DEFW    lenbas-$
          DEFB    249,212,28
          DEFB    126
          DEFB    143,0,18,0,0
lenbas    DEFB    118,0
vars      DEFB    128

last      EQU     $

```